Chapter 5
Unit Commitment

This chapter provides GAMS code for solving unit commitment (UC) problem. The developed GAMS code is linear and is categorized as a MIP model in GAMS. The inputs are generator's characteristics, electricity prices, and demands. The outputs of this code are on/off status of units and their operating schedules.

Every unit commitment problem has three main cost components namely fuel costs, start-up, and shut-down costs. The unit commitment cost calculation is illustrated in Fig. 5.1.

The unit commitment data for ten units are inspired by Ademovic et al. [1] and described in Table 5.1.

5.1 Cost-Based Unit Commitment

5.1.1 Cost Calculation

$$\min_{p^k_{i,t},u_{i,t},y_{i,t},z_{i,t}} \text{OF} = \sum_{i,t} \text{FC}_{i,t} + \text{STC}_{i,t} + \text{SDC}_{i,t} \tag{5.1}$$

The linear version of fuel cost calculation is described in (5.2).

$$0 \le p^k_{i,t} \le \Delta P^k_i u_{i,t}, \forall k = 1 : n \tag{5.2a}$$

$$\Delta P^k_i = \frac{P^{\max}_i - P^{\min}_i}{n} \tag{5.2b}$$

$$P^k_{i,\text{ini}} = (k-1)\Delta P^k_i + P^{\min}_i \tag{5.2c}$$

$$P^k_{i,\text{fin}} = \Delta P^k_i + P^k_{i,\text{ini}} \tag{5.2d}$$

A. Soroudi, *Power System Optimization Modeling in GAMS*,
DOI 10.1007/978-3-319-62350-4_5



Fig. 5.1 Unit commitment cost calculation

$$P_{i,t} = P_i^{\min} u_{i,t} + \sum_k p_{i,t}^k \tag{5.2e}$$

$$C_{i,\text{ini}}^k = a_i \left(P_{i,\text{ini}}^k\right)^2 + b_i P_{i,\text{ini}}^k + c_i \tag{5.2f}$$

$$C_{i,\text{fin}}^k = a_i \left(P_{i,\text{fin}}^k\right)^2 + b_i P_{i,\text{fin}}^k + c_i \tag{5.2g}$$

$$s_i^k = \frac{C_{i,\text{fin}}^k - C_{i,\text{ini}}^k}{\Delta P_i^k} \tag{5.2h}$$

$$\text{FC}_{i,t} = a_i \left(P_i^{\min}\right)^2 + b_i P_i^{\min} + c_i u_{i,t} + \sum_k s_i^k p_{i,t}^k \tag{5.2i}$$

$u_{i,t}$ is the on/off status of the unit i at time t.

5.1.2 Ramp Rate Constraints

The power generation of unit i at time t should be within the operating limits as given in (5.3a). $\underline{P}_{i,t}$ and $\bar{P}_{i,t}$ state the minimum and maximum time-dependent operating limits. These are not necessarily equal to $P_i^{\min}$ and $P_i^{\max}$, respectively.

The upper operating limit ($\bar{P}_{i,t}$) is described in (5.3b) and (5.3c) according to [2]. The ramp up/down constraints are modeled as follows:

$$\underline{P}_{i,t} \le P_{i,t} \le \bar{P}_{i,t} \tag{5.3a}$$

$$\bar{P}_{i,t} \le P_i^{\max} \left[u_{i,t} - z_{i,t+1}\right] + \text{SD}_i z_{i,t+1} \tag{5.3b}$$

$$\bar{P}_{i,t} \le P_{i,t-1} + \text{RU}_i u_{i,t-1} + \text{SU}_i y_{i,t} \tag{5.3c}$$

$$\underline{P}_{i,t} \ge P_{i,t}^{\min} u_{i,t} \tag{5.3d}$$

$$\underline{P}_{i,t} \ge P_{i,t-1} - \text{RD}_i u_{i,t} - \text{SD}_i z_{i,t} \tag{5.3e}$$

Table 5.1 Unit commitment data for ten thermal units

Unit	a_i (\$/MW^2)	b_i (\$/MW)	c_i (\$)	Cd_i (\$)	Cs_i (\$)	RU$_i$ (MW h^{-1})	RD$_i$ (MW h^{-1})	UT$_i$ (h)	DT$_i$ (h)	SD$_i$ (MW h^{-1})	SU$_i$ (MW h^{-1})	$P_i^{\min}$ (MW)	$P_i^{\max}$ $P^{(MW)}$	U_i^0 (h)	$u_{i,t=0}$	S_i^0 (h)
$g1$	0.0148	12.1	82	42.6	42.6	40	40	3	2	90	110	80	200	1	0	1
$g2$	0.0289	12.6	49	50.6	50.6	64	64	4	2	130	140	120	320	2	0	0
$g3$	0.0135	13.2	100	57.1	57.1	30	30	3	2	70	80	50	150	3	0	3
$g4$	0.0127	13.9	105	47.1	47.9	104	104	5	3	240	250	250	520	1	1	0
$g5$	0.0261	13.5	72	56.6	56.9	56	56	4	2	110	130	80	280	1	1	0
$g6$	0.0212	15.4	29	141.5	141.5	30	30	3	2	60	80	50	150	0	0	0
$g7$	0.0382	14	32	113.5	113.5	24	24	3	2	50	60	30	120	0	1	0
$g8$	0.0393	13.5	40	42.6	42.6	22	22	3	2	45	55	30	110	0	0	0
$g9$	0.0396	15	25	50.6	50.6	16	16	0	0	35	45	20	80	0	0	0
$g10$	0.0510	14.3	15	57.1	57.1	12	12	0	0	30	40	20	60	0	0	0

In order to explain the upper operating limits described in (5.3) some assumptions should be taken into account:

- It should be always less than capacity of unit i, this means that $\bar{P}_{i,t} \leq P_i^{\max}$.
- In case of unit shut-down in the next hour $(t+1)$: $\bar{P}_{i,t} \leq \mathrm{SD}_i z_{i,t+1}$. Since $P_{i,t+1} = 0$ so $P_{i,t}$ cannot be more than SD_i.
- If the unit has been on in the previous hour $(u_{i,t-1} = 1)$ and is going to remain on then $P_{i,t}$ cannot be increased more than RU_i. This means that $\bar{P}_{i,t} \leq P_{i,t-1} + \mathrm{RU}_i u_{i,t-1}$.
- If the unit has been off in the previous hour $(u_{i,t-1} = 0)$ and it is turned on at time t $(y_{i,t} = 1)$ then $P_{i,t}$ cannot be more than SU_i. This means that $\bar{P}_{i,t} \leq \mathrm{SU}_i y_{i,t}$.

The combination of all these cases is enforced by (5.3b) and (5.3c).

In order to explain the lower operating limits described in (5.3) some assumptions should be taken into account:

- If the unit is on at time t then the generated power should be greater than $P_{i,t}^{\min} u_{i,t}$.
- If the unit is on at time $t-1$ and remains to be on at time $t, t+1$ then the generated power at time t should be greater than $P_{i,t-1} - \mathrm{RD}_i u_{i,t}$.
- If the unit is on at time $t-1$ and turned off at time t then the generated power at time $t-1$ should be $P_{i,t-1} \leq \mathrm{SD}_i z_{i,t}$.

The combination of all these cases is enforced by (5.3d) and (5.3e).

5.1.3 Min Up/Down Time Constraints

The on/off states of unit i at time t are described by $u_{i,t}$. Additionally, start-up/shut-down are given by $y_{i,t}, z_{i,t}$ in (5.4).

$$y_{i,t} - z_{i,t} = u_{i,t} - u_{i,t-1} \tag{5.4a}$$

$$y_{i,t} + z_{i,t} \leq 1 \tag{5.4b}$$

$$y_{i,t}, z_{i,t}, u_{i,t} \in \{0, 1\}$$

The minimum up time (UT_i) of unit i is modeled in (5.5) as proposed in [2].

$$\sum_{t=1}^{\zeta_i} 1 - u_{i,t} = 0 \tag{5.5a}$$

$$\sum_{t=k}^{k+\mathrm{UT}_i-1} u_{i,t} \geq \mathrm{UT}_i y_{i,k}, \forall k = \zeta_i + 1 \ldots T - \mathrm{UT}_i + 1 \tag{5.5b}$$

$$\sum_{t=k}^{T} u_{i,t} - y_{i,t} \geq 0, \forall k = T - \mathrm{UT}_i + 2 \ldots T \tag{5.5c}$$

$$\zeta_i = \min\left\{T, \left(\mathrm{UT}_i - U_i^0\right) u_{i,t=0}\right\} \tag{5.5d}$$

The minimum up time (DT_i) of unit i is modeled in (5.6) as proposed in [2].

$$\sum_{t=1}^{\xi_i} u_{i,t} = 0 \tag{5.6a}$$

$$\sum_{t=k}^{k+\mathrm{DT}_i-1} 1 - u_{i,t} \geq \mathrm{DT}_i z_{i,k}, \forall k = \xi_i + 1 \ldots T - \mathrm{DT}_i + 1 \tag{5.6b}$$

$$\sum_{t=k}^{T} 1 - u_{i,t} - z_{i,t} \geq 0, \forall k = T - \mathrm{DT}_i + 2 \ldots T \tag{5.6c}$$

$$\xi_i = \min\left\{T, \left(\mathrm{DT}_i - S_i^0\right)[1 - u_{i,t=0}]\right\} \tag{5.6d}$$

The realistic start-up and shut-down costs depend on how long the unit has been off or on, receptively. Some references have provided the detailed formulation for modeling these cost terms. Here for simplicity, the start-up (Sd_i) and shut-down (Sd_i) costs are modeled as constant values as (5.7).

$$\mathrm{STC}_{i,t} = Cs_i y_{i,t} \tag{5.7a}$$

$$\mathrm{SDC}_{i,t} = Sd_i z_{i,t} \tag{5.7b}$$

5.1.4 Demand-Generation Balance

In cost-based UC, the hourly total generation should be equal to the hourly demand:

$$\sum_i P_{i,t} \geq L_t \tag{5.8}$$

The hourly demand and price values vs time are depicted in Fig. 5.2.

The GAMS code for solving the cost-based unit commitment is provided in GCode 5.1.

GCode 5.1 Cost-based unit commitment example for ten unit system

```
Sets t    time /t1*t24/,
          i generators / g1*g10 /,
                                        k cost segments /sg1*sg20/,
          char /ch1*ch2/;
Alias (t,h);
Table gendata(i,*) generator cost characteristics and limits
    a      b     c     CostsD   costst  RU  RD  UT DT SD   SU   Pmin Pmax U0 Uini S0

g1  0.014  12.1  82    42.6     42.6    40  40  3  2  90   110  80   200  1  0    1
g2  0.028  12.6  49    50.6     50.6    64  64  4  2  130  140  120  320  2  0    0
g3  0.013  13.2  100   57.1     57.1    30  30  3  2  70   80   50   150  3  0    3
g4  0.012  13.9  105   47.1     47.9    104 104 5  3  240  250  250  520  1  1    0
g5  0.026  13.5  72    56.6     56.9    56  56  4  2  110  130  80   280  1  1    0
```

```
g6   0.021  15.4   29    141.5   141.5    30   30   3   2   60    80    50    150   0   0     0
g7   0.038  14.0   32    113.5   113.5    24   24   3   2   50    60    30    120   0   1     0
g8   0.039  13.5   40    42.6     42.6    22   22   3   2   45    55    30    110   0   0     0
g9   0.039  15.0   25    50.6     50.6    16   16   0   0   35    45    20    80    0   0     0
g10  0.051  14.3   15    57.1     57.1    12   12   0   0   30    40    20    60    0   0     0;
Parameter data(k,i,*);
data(k,i,'DP')=(gendata(i,"Pmax")-gendata(i,"Pmin"))/card(k);
data(k,i,'Pini')= (ord(k)-1)*data(k,i,'DP')+gendata(i,"Pmin");
data(k,i,'Pfin')=data(k,i,'Pini')+data(k,i,'DP');
data(k,i,'Cini')=gendata(i,"a")*power(data(k,i,'Pini'),2)
+gendata(i,"b")*data(k,i,'Pini')+gendata(i,"c");
data(k,i,'Cfin')=gendata(i,"a")*power(data(k,i,'Pfin'),2)
+gendata(i,"b")*data(k,i,'Pfin')+gendata(i,"c");
data(k,i,'s')= (data(k,i,'Cfin')-data(k,i,'Cini'))/data(k,i,'DP');
gendata(i,'Mincost')=gendata(i,'a')*power(gendata(i,"Pmin"),2)
+gendata(i,'b')*gendata(i,"Pmin")+gendata(i,'c');
Table dataLP(t,*)
          lambda      load
t1        14.72       883
t2        15.62       915
t23       20.50       915
t24       15.62       834;
Parameter unit(i,char);
unit(i,'ch1')=24;
unit(i,'ch2')=(gendata(i,'UT')-gendata(i,'U0'))*gendata(i,'Uini');
parameter unit2(i,char); unit2(i,'ch1')=24;
unit2(i,'ch2')=(gendata(i,'DT')-gendata(i,'S0'))*(1-gendata(i,'Uini'));
gendata(i,'Lj')=smin(char,unit(i,char)); gendata(i,'Fj')=smin(char,unit2(i,char));
variable costThermal;positive variables pu(i,t),p(i,t),StC(i,t),SDC(i,t),Pk(i,t,k);
Binary variable u(i,t),y(i,t),z(i,t);
p.up(i,t) = gendata(i,"Pmax") ; p.lo(i,t) = 0; Pk.up(i,t,k)=data(k,i,'DP');
Pk.lo(i,t,k)=0; p.up(i,t) = gendata(i,"Pmax"); pu.up(i,h) = gendata(i,"Pmax") ;
Equations Uptime1,Uptime2,Uptime3,Dntime1,Dntime2,Dntime3,Ramp1,Ramp2,
 Ramp3,Ramp4,startc,shtdnc,genconst1,Genconst2,Genconst3,Genconst4,balance;
Uptime1(i)$(gendata(i,"Lj")>0) ..
sum(t$(ord(t)< (gendata(i,"Lj")+1)),1-U(i,t))=e=0;
Uptime2(i)$(gendata(i,"UT")>1) ..
sum(t$(ord(t)>24-gendata(i,"UT")+1),U(i,t)-y(i,t))=g=0;
Uptime3(i,t)$(ord(t)>gendata(i,"Lj") and ord(t)<24-gendata(i,"UT")+2 and
not(gendata(i,"Lj")>24-gendata(i,"UT"))) .. sum(h$((ord(h)>ord(t)-1) and
(ord(h)<ord(t)+gendata(i,"UT"))),U(i,h)) =g=gendata(i,"UT")*y(i,t);
Dntime1(i)$(gendata(i,"Fj")>0) .. sum(t$(ord(t)< (gendata(i,"Fj")+1)),U(i,t))=e=0;
Dntime2(i)$(gendata(i,"DT")>1) ..
sum(t$(ord(t)>24-gendata(i,"DT")+1),1-U(i,t)-z(i,t))=g=0;
Dntime3(i,t)$(ord(t)>gendata(i,"Fj") and ord(t)<24-gendata(i,"DT")+2
and not(gendata(i,"Fj")>24-gendata(i,"DT"))   ) .. sum(h$((ord(h)>ord(t)-1)
and (ord(h)<ord(t)+gendata(i,"DT"))),1-U(i,h))=g=gendata(i,"DT")*z(i,t);
startc(i,t) .. StC(i,t)=g=gendata(i,"costst")*y(i,t);
shtdnc(i,t) .. SDC(i,t)=g=gendata(i,"CostsD")*z(i,t);
Genconst1(i,h)              .. p(i,h)=e=u(i,h)*gendata(i,"Pmin")+sum(k,Pk(i,h,k));
Genconst2(i,h)$(ord(h)>0)..U(i,h)=e=U(i,h-1)$(ord(h)>1)+gendata(i,"Uini")$(ord(h)
      =1)
                                         +y(i,h)-z(i,h);
Genconst3(i,t,k)            .. Pk(i,t,k)=l=U(i,t)*data(k,i,'DP');
Genconst4 .. costThermal=e=sum((i,t),StC(i,t)+SDC(i,t))+sum((t,i),
u(i,t)*gendata(i,'Mincost')+sum(k,data(k,i,'s')*pk(i,t,k)));
Ramp1(i,t) .. p(i,t-1)-p(i,t)=l=U(i,t)*gendata(i,'RD')+z(i,t)*gendata(i,"SD");
Ramp2(i,t) .. p(i,t)=l=pu(i,t);
Ramp3(i,t)$(ord(t)<24)..pu(i,t)=l=(u(i,t)-z(i,t+1))*gendata(i,"Pmax")
                                                      +z(i,t+1)*gendata(i,"SD");
Ramp4(i,t)$(ord(t)>1)..pu(i,t)=l=p(i,t-1)+U(i,t-1)*gendata(i,'RU')
                                                        +y(i,t)*gendata(i,"SU");
Balance(t)..        sum(i,p(i,t))=e= dataLP(t,'load');
Model UCLP /all/;
Option optcr=0.0;
Solve UCLP minimizing costThermal using mip ;
```



Fig. 5.2 Hourly demand and price values vs time



Fig. 5.3 Power schedules in cost-based unit commitment vs time

The total operating costs are \$485,240.189. The start-up, shut-down, and fuel costs are \$442.100, \$0, and \$484,798.089, respectively. The power schedules of thermal units in cost-based unit commitment are depicted in Fig. 5.3.

5.2 Cost-Based UC with Additional Constraints

In order to improve the security and efficiency of the energy supply, some additional constraints are needed to be considered in UC formulation. Some of these constraints are discussed and formulated in this section.

5.2.1 Cost-Based UC with Reserve Constraints

One of the most important resources which is used by the system operator is called spinning reserve (SR) [3]. It is used to cover the sudden increase in demand, rapid reduction of renewable energy production, or unplanned generating unit outage. The SR is supplied by online generating units which are synchronized to the system and are able to ramp-up in order to meet the demand. The UC-reserve constrained formulation is provided as follows:

$$\min_{p^k_{i,t},u_{i,t},y_{i,t},z_{i,t}} \quad \mathrm{OF} = \sum_{i,t} \mathrm{FC}_{i,t} + \mathrm{STC}_{i,t} + \mathrm{SDC}_{i,t} \tag{5.9a}$$

$$\sum_i P_{i,t} \geq L_t \tag{5.9b}$$

$$R_{i,t} \leq \bar{P}_{i,t} - P_{i,t} \tag{5.9c}$$

$$\sum_i R_{i,t} \geq \gamma L_t \tag{5.9d}$$

Subject to:

(5.2), (5.3), (5.4), (5.5), (5.6), (5.7)

where $R_{i,t}$ is the reserve provided by online unit i at time t. γ is the percentage of demand which specifies the reserve requirement. It is usually expressed as a percentage of demand at time t. The GAMS code for solving the (5.9) is described in GCode 5.2. It should be noted that the generating unit characteristics are unchanged but the hourly demand values specified in GCode 5.1 and Fig. 5.2 are reduced by 55%. The γ value is specified by the system operator and is dependent on the system specification. Here, it is assumed to be 40% for simulation purpose. The total operating costs without the reserve constraint is \$2.1039 $\times$ 10^5. However, if the reserve constraint is taken into account then the operating costs would increase to \$2.1090 $\times$ 10^5. The operating schedules of thermal units considering reserve constraint are provided in Table 5.2.

GCode 5.2 Cost-based unit commitment with reserve constraint

```
Sets t /t1*t24/, i /g1*g10/, k /sg1*sg20/,char /ch1*ch2/;alias(t,h);alias(i,g);
Table gendata(i,*) generator cost characteristics and limits
* Removed for saving space ;
Parameter data(k,i,*);
data(k,i,'DP')=(gendata(i,"Pmax")-gendata(i,"Pmin"))/card(k);
data(k,i,'Pini')= (ord(k)-1)*data(k,i,'DP')+gendata(i,"Pmin");
data(k,i,'Pfin')=data(k,i,'Pini')+data(k,i,'DP');
data(k,i,'Cini')=gendata(i,"a")*power(data(k,i,'Pini'),2)
+gendata(i,"b")*data(k,i,'Pini')+gendata(i,"c");
```

```
data(k,i,'Cfin')=gendata(i,"a")*power(data(k,i,'Pfin'),2)
+gendata(i,"b")*data(k,i,'Pfin')+gendata(i,"c");
data(k,i,'s')= (data(k,i,'Cfin')-data(k,i,'Cini'))/data(k,i,'DP');
gendata(i,'Mincost')=gendata(i,'a')*power(gendata(i,"Pmin"),2)
+gendata(i,'b')*gendata(i,"Pmin")+gendata(i,'c');
table dataLP(t,*)
          lambda       load
* Removed for saving space     ;
dataLP(t,'load')=dataLP(t,'load')*0.45; Parameter unit(i,char);
unit(i,'ch1')=24; unit(i,'ch2')=(gendata(i,'UT')-gendata(i,'U0'))*gendata(i,'Uini')
     ;
Parameter unit2(i,char); unit2(i,'ch1')=24;
unit2(i,'ch2')=(gendata(i,'DT')-gendata(i,'S0'))*(1-gendata(i,'Uini'));
gendata(i,'Lj')=smin(char,unit(i,char)); gendata(i,'Fj')=smin(char,unit2(i,char));
Variable  costThermal; Binary variable u(i,t),y(i,t),z(i,t);
Positive variables pu(i,t),p(i,t),StC(i,t),SDC(i,t),Pk(i,t,k);
p.up(i,t) = gendata(i,"Pmax") ; p.lo(i,t) =0;
Pk.up(i,t,k)=data(k,i,'DP'); Pk.lo(i,t,k)=0;
p.up(i,t) = gendata(i,"Pmax"); pu.up(i,h) = gendata(i,"Pmax");
Equations Uptime1,Uptime2,Uptime3,Dntime1,Dntime2,Dntime3,
Ramp1,Ramp2,Ramp3,Ramp4,startc,shtdnc,genconst1,
Genconst2,Genconst3,Genconst4,balance,reserve;
Uptime1(i)$(gendata(i,"Lj")>0)..
 sum(t$(ord(t)< (gendata(i,"Lj")+1)),1-U(i,t))=e=0;
Uptime2(i)$(gendata(i,"UT")>1)..
 sum(t$(ord(t)>24-gendata(i,"UT")+1),U(i,t)-y(i,t))=g=0;
Uptime3(i,t)$(ord(t)>gendata(i,"Lj") and ord(t)<24-gendata(i,"UT")+2 and
not(gendata(i,"Lj")>24-gendata(i,"UT")))
      .. sum(h$((ord(h)>ord(t)-1) and (ord(h)<ord(t)+gendata(i,"UT"))),U(i,h)) =g=
              gendata(i,"UT")*y(i,t);
Dntime1(i)$(gendata(i,"Fj")>0) ..    sum(t$(ord(t)< (gendata(i,"Fj")+1)),U(i,t))=e
     =0;
Dntime2(i)$(gendata(i,"DT")>1) ..
                              sum(t$(ord(t)>24-gendata(i,"DT")+1),1-U(i,t)-z(i,t))=g
                                   =0;
Dntime3(i,t)$(ord(t)>gendata(i,"Fj") and ord(t)<24-gendata(i,"DT")+2 and
not(gendata(i,"Fj")>24-gendata(i,"DT")))
.. sum(h$((ord(h)>ord(t)-1) and (ord(h)<ord(t)+gendata(i,"DT"))),1-U(i,h)) =g=
gendata(i,"DT")*z(i,t);
startc(i,t) .. StC(i,t)=g=gendata(i,"costst")*y(i,t);
shtdnc(i,t) .. SDC(i,t)=g=gendata(i,"CostsD")*z(i,t);
genconst1(i,h) .. p(i,h)=e=u(i,h)*gendata(i,"Pmin")+sum(k,Pk(i,h,k));
Genconst2(i,h)$(ord(h)>0)..U(i,h)=e=
U(i,h-1)$(ord(h)>1)+gendata(i,"Uini")$(ord(h)=1)+y(i,h)-z(i,h);
Genconst3(i,t,k) .. Pk(i,t,k)=l=U(i,t)*data(k,i,'DP');
Genconst4  .. costThermal=e=sum((i,t),StC(i,t)+SDC(i,t))+sum((t,i),
                    u(i,t)*gendata(i,'Mincost')+sum(k,data(k,i,'s')*pk(i,t,k)));
Ramp1(i,t) .. p(i,t-1)-p(i,t)=l=U(i,t)*gendata(i,'RD')+z(i,t)*gendata(i,"SD");
Ramp2(i,t) .. p(i,t)=l=pu(i,t);
Ramp3(i,t)$(ord(t)<24)..pu(i,t)=l=(u(i,t)-z(i,t+1))*gendata(i,"Pmax")
                                                  +z(i,t+1)*gendata(i,"SD");
Ramp4(i,t)$(ord(t)>1)  .. pu(i,t)=l=p(i,t-1)+U(i,t-1)*gendata(i,'RU')
                                                +y(i,t)*gendata(i,"SU");
balance(t) .. sum(i,p(i,t))=e= dataLP(t,'load');
reserve(t) .. sum(i,pu(i,t)-p(i,t))=g=0.40*dataLP(t,'load');
Model UCLP /all/; Option optcr=0.0;
Solve UCLP minimizing costThermal using mip ;
```

Table 5.2 Power schedules ($P_{i,t}$) of thermal units considering reserve constraint

Time	$g1$	$g3$	$g4$	$g5$	$g7$	$g8$	$g10$
t_1			250.0	90.4	57.0		
t_2			250.0	107.8	54.0		
t_3	94.5		250.0	80.0	30.0		
t_4	92.1	65.0	250.0	80.0	30.0		
t_5	104.0	72.2	250.0	80.0	30.0		20.0
t_6	110.0	79.0	250.0	80.0	30.0	30.0	20.0
t_7	123.7	95.0	250.0	80.0	30.0	30.0	20.0
t_8	128.0	100.6	250.0	80.0	30.0	30.0	20.0
t_9	134.0	106.8	250.0	80.0	30.0	34.0	20.0
t_{10}	134.0	106.8	250.0	80.0	30.0	34.0	20.0
t_{11}	133.5	105.0	250.0	80.0	30.0	30.0	20.0
t_{12}	134.6	110.0	250.0	80.0	30.0	34.0	
t_{13}	133.7	105.0	250.0	80.0	30.0	30.0	
t_{14}	122.0	90.6	250.0	80.0	30.0	30.0	
t_{15}	128.0	97.6	250.0	80.0	30.0	30.0	
t_{16}	122.0	90.6	250.0	80.0	30.0	30.0	
t_{17}	98.0	68.2	250.0	80.0	30.0	30.0	
t_{18}	110.0	77.3	250.0		30.0	30.0	
t_{19}	104.2	52.9	250.0		30.0	30.0	
t_{20}	121.6		250.0		30.0	30.0	
t_{21}	104.9		250.0		30.0	30.0	
t_{22}	88.3		250.0		30.0	30.0	
t_{23}	101.8		250.0		30.0	30.0	
t_{24}	95.3		250.0		30.0		

The reserve provision ($R_{i,t}$) by thermal units considering reserve constraint is given in Table 5.3.

5.2.2 *Cost-Based UC Considering Generator Contingency*

In this section, the cost-based UC is formulated in order to consider the generator outage contingencies. The system operator is willing to be robust against the unexpected outages of generating units. It means that in case of any generating unit outage the reserve resource by the remaining units should be able to supply the lost generating unit. This constraint is modeled as follows:

$$\min_{p^k_{i,t},u_{i,t},y_{i,t},z_{i,t}} \mathrm{OF} = \sum_{i,t} \mathrm{FC}_{i,t} + \mathrm{STC}_{i,t} + \mathrm{SDC}_{i,t} \tag{5.10a}$$

Table 5.3 Reserve provision ($R_{i,t}$) by thermal units considering reserve constraint

Time	$g1$	$g3$	$g4$	$g5$	$g7$	$g8$	$g10$
t_1			270.0	189.7	63.0		
t_2			99.1	38.6	27.0		
t_3	15.5		34.6	83.8	48.0		
t_4	42.5	15.0	69.4	56.0	24.0		
t_5	28.1	22.8	71.6	56.0	24.0		20.0
t_6	34.0		104.0	56.0	8.6	25.0	12.0
t_7	26.4	14.0	97.2	56.0	24.0	22.0	12.0
t_8	35.7	24.5	81.3	56.0	24.0	22.0	12.0
t_9	34.0	23.8	94.1	56.0	24.0	18.0	12.0
t_{10}	40.0	30.0	77.9	56.0	24.0	22.0	12.0
t_{11}	40.6	31.8	71.1	56.0	24.0	26.0	10.0
t_{12}	38.9	25.0	93.5	56.0	24.0	18.0	
t_{13}	40.9	35.0	69.6	56.0	24.0	26.0	
t_{14}	51.7	44.5	42.9	56.0	24.0	22.0	
t_{15}	34.0	23.0	87.3	56.0	24.0	22.0	
t_{16}	46.0	37.1	56.0	56.0	24.0	22.0	
t_{17}	42.5		104.0	30.0	24.0	22.0	
t_{18}	28.0	21.0	104.0		24.0	22.0	
t_{19}	45.8	17.1	77.9		24.0	22.0	
t_{20}	22.6		104.0		24.0	22.0	
t_{21}	56.7		63.3		24.0	22.0	
t_{22}	56.7		56.7		24.0	22.0	
t_{23}	21.7		104.0		24.0	15.0	
t_{24}			98.1			52.0	

$$\sum_i P_{i,t} \geq L_t \tag{5.10b}$$

$$R_{i,t} \leq \bar{P}_{i,t} - P_{i,t} \tag{5.10c}$$

$$\sum_{i \neq i'} R_{i,t} \geq P_{i',t} \quad \forall i' \in \Omega_c \tag{5.10d}$$

Subject to:

(5.2), (5.3), (5.4), (5.5), (5.6), (5.7)

where Ω_c is the set of contingencies for generating units. The GAMS code for solving the (5.10) is described in GCode 5.3. As previously stated, the total operating costs without the generator contingency constraint are \$$2.1039 \times 10^5$. However, if the generator contingency constraint (the outage of all generating units except $g4$ is considered) is taken into account then the operating costs would increase to \$$2.1041 \times 10^5$.

GCode 5.3 Cost-based unit commitment considering generator contingencies

```
Sets t    time /t1*t24/
      i    generators indices / g1*g10 /, k   cost segments /sg1*sg20/
      char /ch1*ch2/, g(i) /g1*g3,g5*g10/; Alias (t,h);
table gendata(i,*) generator cost characteristics and limits
   a  b  c CostsD  costst  RU RD UT DT SD SU Pmin Pmax U0 Uini S0
* Removed for saving space           ;
parameter data(k,i,*);
data(k,i,'DP')=(gendata(i,"Pmax")-gendata(i,"Pmin"))/card(k);
data(k,i,'Pini')= (ord(k)-1)*data(k,i,'DP')+gendata(i,"Pmin");
data(k,i,'Pfin')=data(k,i,'Pini')+data(k,i,'DP');
data(k,i,'Cini')=gendata(i,"a")*power(data(k,i,'Pini'),2)
+gendata(i,"b")*data(k,i,'Pini')+gendata(i,"c");
data(k,i,'Cfin')=gendata(i,"a")*power(data(k,i,'Pfin'),2)
+gendata(i,"b")*data(k,i,'Pfin')+gendata(i,"c");
data(k,i,'s')= (data(k,i,'Cfin')-data(k,i,'Cini'))/data(k,i,'DP');
gendata(i,'Mincost')=gendata(i,'a')*power(gendata(i,"Pmin"),2)
+gendata(i,'b')*gendata(i,"Pmin")+gendata(i,'c');
Table dataLP(t,*)
           lambda        load
* Removed for saving space;
dataLP(t,'load')=dataLP(t,'load')*0.45;
Parameter unit(i,char); unit(i,'ch1')=24;
unit(i,'ch2')=(gendata(i,'UT')-gendata(i,'U0'))*gendata(i,'Uini');
parameter unit2(i,char); unit2(i,'ch1')=24;
unit2(i,'ch2')=(gendata(i,'DT')-gendata(i,'S0'))*(1-gendata(i,'Uini'));
gendata(i,'Lj')=smin(char,unit(i,char));gendata(i,'Fj')=smin(char,unit2(i,char));
Variable  costThermal   ;
positive variables pu(i,t),p(i,t),StC(i,t),SDC(i,t),Pk(i,t,k);
Binary variable u(i,t),y(i,t),z(i,t);
p.up(i,t) = gendata(i,"Pmax") ; p.lo(i,t) = 0;
Pk.up(i,t,k)=data(k,i,'DP'); Pk.lo(i,t,k)=0;
p.up(i,t) = gendata(i,"Pmax"); pu.up(i,h) = gendata(i,"Pmax");
Equations Uptime1,Uptime2,Uptime3,Dntime1,Dntime2,Dntime3,Ramp1,Ramp2,Ramp3,Ramp4
,startc,shtdnc,genconst1,Genconst2  ,Genconst3,Genconst4,balance,reserve;
Uptime1(i)$(gendata(i,"Lj")>0)..sum(t$(ord(t)< (gendata(i,"Lj")+1)),1-U(i,t))=e=0;
Uptime2(i)$(gendata(i,"UT")>1)
                            ..sum(t$(ord(t)>24-gendata(i,"UT")+1),U(i,t)-y(i,t))=g=0;
Uptime3(i,t)$(ord(t)>gendata(i,"Lj") and ord(t)<24-gendata(i,"UT")+2
and not(gendata(i,"Lj")>24-gendata(i,"UT"))    ) .. sum(h$((ord(h)>ord(t)-1) and
(ord(h)<ord(t)+gendata(i,"UT"))),U(i,h)) =g=gendata(i,"UT")*y(i,t);
Dntime1(i)$(gendata(i,"Fj")>0)..sum(t$(ord(t)<(gendata(i,"Fj")+1)),U(i,t))=e=0;
Dntime2(i)$(gendata(i,"DT")>1)
.. sum(t$(ord(t)>24-gendata(i,"DT")+1),1-U(i,t)-z(i,t))=g=0;
Dntime3(i,t)$(ord(t)>gendata(i,"Fj") and ord(t)<24-gendata(i,"DT")+2 and
not(gendata(i,"Fj")>24-gendata(i,"DT"))    ) .. sum(h$((ord(h)>ord(t)-1) and
(ord(h)<ord(t)+gendata(i,"DT"))),1-U(i,h)) =g=gendata(i,"DT")*z(i,t);
startc(i,t) .. StC(i,t)=g=gendata(i,"costst")*y(i,t);
shtdnc(i,t) .. SDC(i,t)=g=gendata(i,"CostsD")*z(i,t);
genconst1(i,h) .. p(i,h)=e=u(i,h)*gendata(i,"Pmin")+sum(k,Pk(i,h,k));
Genconst2(i,h)$(ord(h)>0) .. U(i,h)=e=U(i,h-1)$(ord(h)>1)
+gendata(i,"Uini")$(ord(h)=1)+y(i,h)-z(i,h);
Genconst3(i,t,k) .. Pk(i,t,k)=l=U(i,t)*data(k,i,'DP');
Genconst4  ..costThermal=e=sum((i,t),StC(i,t)+SDC(i,t))
+sum((t,i),u(i,t)*gendata(i,'Mincost')+sum(k,data(k,i,'s')*pk(i,t,k)));
Ramp1(i,t).. p(i,t-1)-p(i,t)=l=U(i,t)*gendata(i,'RD')+z(i,t)*gendata(i,"SD");
Ramp2(i,t).. p(i,t)=l=pu(i,t);
Ramp3(i,t)$(ord(t)<24)..pu(i,t)=l=(u(i,t)-z(i,t+1))*gendata(i,"Pmax")
                                                   +z(i,t+1)*gendata(i,"SD");
Ramp4(i,t)$(ord(t)>1) ..pu(i,t)=l=p(i,t-1)+U(i,t-1)*gendata(i,'RU')
                                                     +y(i,t)*gendata(i,"SU");
balance(t) ..       sum(i,p(i,t))=e= dataLP(t,'load');
```

```
reserve(t,g) .. sum(i$(ord(i) <> ord(g) ),pu(i,t)-p(i,t))=g=1*p(g,t);
Model UCLP /all/; Option optcr=0.0;
Solve UCLP minimizing costThermal using mip ;
```

Power schedules ($P_{i,t}$) by thermal units considering generator contingency constraint are provided in Table 5.4.

The reserve provision ($R_{i,t}$) by thermal units considering generator contingencies is provided in Table 5.5.

5.2.3 *Cost-Based UC with Demand Flexibility Constraints*

The balance between generation and demand is traditionally managed by scheduling the generating units. However, this paradigm is changing gradually. In other words, the demand values can also change intentionally to increase the efficiency of UC

Table 5.4 Power schedules ($P_{i,t}$) by thermal units considering generator contingency constraint

Time	$g1$	$g3$	$g4$	$g5$	$g7$	$g8$	$g10$
t_1	0.0	0.0	250.0	93.4	54.0	0.0	0.0
t_2	0.0	51.8	250.0	80.0	30.0	0.0	0.0
t_3	0.0	74.5	250.0	80.0	30.0	0.0	20.0
t_4	110.0	104.5	250.0	0.0	30.6	0.0	22.0
t_5	128.0	98.2	250.0	0.0	30.0	30.0	20.0
t_6	146.0	115.0	250.0	0.0	30.0	38.0	20.0
t_7	152.0	130.0	250.0	0.0	34.5	40.2	22.0
t_8	158.0	130.1	250.0	0.0	34.5	42.0	24.0
t_9	162.8	135.0	250.0	0.0	39.0	42.0	26.0
t_{10}	162.8	135.0	250.0	0.0	39.0	42.0	26.0
t_{11}	158.0	135.0	250.0	0.0	39.0	42.0	24.5
t_{12}	158.0	130.1	250.0	0.0	34.5	42.0	24.0
t_{13}	152.0	130.0	250.0	0.0	34.5	40.2	22.0
t_{14}	146.0	118.6	250.0	0.0	30.0	38.0	20.0
t_{15}	146.1	125.0	250.0	0.0	34.5	38.0	22.0
t_{16}	146.0	118.6	250.0	0.0	30.0	38.0	20.0
t_{17}	139.0	113.3	250.0	0.0	0.0	34.0	20.0
t_{18}	137.3	110.0	250.0	0.0	0.0	0.0	0.0
t_{19}	122.1	95.0	250.0	0.0	0.0	0.0	0.0
t_{20}	106.6	75.0	250.0	0.0	0.0	0.0	0.0
t_{21}	98.0	66.9	250.0	0.0	0.0	0.0	0.0
t_{22}	92.0	56.3	250.0	0.0	0.0	0.0	0.0
t_{23}	96.8	65.0	250.0	0.0	0.0	0.0	0.0
t_{24}	125.3	0.0	250.0	0.0	0.0	0.0	0.0

Table 5.5 Reserve provision ($R_{i,t}$) by thermal units considering generator contingency constraint

Time	$g1$	$g3$	$g4$	$g5$	$g7$	$g8$	$g10$
t_1	0.0	0.0	270.0	186.7	66.0	0.0	0.0
t_2	0.0	28.3	0.0	69.4	48.0	0.0	0.0
t_3	0.0	7.3	1.8	30.0	22.8	0.0	20.0
t_4	0.0	0.0	76.6	0.0	23.5	0.0	10.0
t_5	22.0	0.0	64.5	0.0	24.6	25.0	14.0
t_6	0.0	4.0	104.0	0.0	24.0	14.0	0.0
t_7	0.0	8.7	104.0	0.0	19.5	19.9	0.0
t_8	0.0	0.0	103.9	0.0	24.0	20.2	10.0
t_9	0.0	15.0	96.3	0.0	19.5	22.0	10.0
t_{10}	0.0	15.0	89.8	0.0	24.0	22.0	12.0
t_{11}	0.0	0.0	98.5	0.0	24.0	22.0	13.6
t_{12}	0.0	0.0	95.1	0.0	28.5	22.0	12.5
t_{13}	0.0	0.2	104.0	0.0	24.0	23.9	0.0
t_{14}	0.0	3.9	104.0	0.0	0.0	24.2	14.0
t_{15}	0.0	21.1	73.5	0.0	19.5	22.0	10.0
t_{16}	0.0	0.0	90.0	0.0	20.0	22.0	14.0
t_{17}	0.0	14.0	104.0	0.0	0.0	11.0	10.0
t_{18}	6.0	33.3	104.0	0.0	0.0	0.0	0.0
t_{19}	0.0	18.1	104.0	0.0	0.0	0.0	0.0
t_{20}	0.0	31.6	75.0	0.0	0.0	0.0	0.0
t_{21}	0.0	31.1	66.9	0.0	0.0	0.0	0.0
t_{22}	0.0	0.0	92.0	0.0	0.0	0.0	0.0
t_{23}	0.0	5.0	91.8	0.0	0.0	0.0	0.0
t_{24}	0.0	95.0	30.3	0.0	0.0	0.0	0.0

problems. In this modern context, the demand values are no longer strict and can be dispatched. The UC problem incorporating a simple demand response model is presented as follows:

$$\min_{p^k_{i,t}, u_{i,t}, y_{i,t}, z_{i,t}} \quad \mathrm{OF} = \sum_{i,t} \mathrm{FC}_{i,t} + \mathrm{STC}_{i,t} + \mathrm{SDC}_{i,t} \tag{5.11a}$$

$$\sum_{i} P_{i,t} \geq D_t \tag{5.11b}$$

$$(1 - \varsigma_{\min}) L_t \leq D_t \leq (1 + \varsigma_{\max}) L_t \tag{5.11c}$$

$$\sum_{t} D_t = \sum_{t} L_t \tag{5.11d}$$

Subject to:

(5.2), (5.3), (5.4), (5.5), (5.6), (5.7)

Equation (5.11c) states the demand variation ranges. $\varsigma_{\min/\max}$ determine the min/-max flexibility of demand response. Equation (5.11d) states that the total energy of the consumer does not change over the operating horizon. The $\varsigma_{\min/\max}$ are assumed to be 10%.

The GAMS code for solving the (5.11) is described in GCode 5.4. As previously stated, the total operating costs without the generator contingency constraint are $\$2.1039 \times 10^5$. However, if the demand response flexibility is available then the operating costs would decrease to $\$2.0965 \times 10^5$. The power schedules ($P_{i,t}$) by thermal units considering demand response constraint are given in Table 5.6.

GCode 5.4 Cost-based unit commitment considering demand response

```
Sets t/t1*t24/, i/g1*g10 /,k/sg1*sg20/,char /ch1*ch2/;alias (t,h);
Table gendata(i,*) generator cost characteristics and limits
     a  b  c    CostsD   costst  RU RD UT DT SD SU Pmin Pmax U0 Uini S0
* Removed for saving space       ; parameter data(k,i,*);
data(k,i,'DP')=(gendata(i,"Pmax")-gendata(i,"Pmin"))/card(k);
data(k,i,'Pini')= (ord(k)-1)*data(k,i,'DP')+gendata(i,"Pmin");
data(k,i,'Pfin')=data(k,i,'Pini')+data(k,i,'DP');
data(k,i,'Cini')=gendata(i,"a")*power(data(k,i,'Pini'),2)
         +gendata(i,"b")*data(k,i,'Pini')+gendata(i,"c");
data(k,i,'Cfin')=gendata(i,"a")*power(data(k,i,'Pfin'),2)
+gendata(i,"b")*data(k,i,'Pfin')+gendata(i,"c");
data(k,i,'s')= (data(k,i,'Cfin')-data(k,i,'Cini'))/data(k,i,'DP');
gendata(i,'Mincost')=gendata(i,'a')*power(gendata(i,"Pmin"),2)
                     +gendata(i,'b')*gendata(i,"Pmin")+gendata(i,'c');
Table dataLP(t,*)
          lambda        load
* Removed for saving space       ;
dataLP(t,'load')=dataLP(t,'load')*0.45;
parameter unit(i,char);
unit(i,'ch1')=24; unit(i,'ch2')=
(gendata(i,'UT')-gendata(i,'U0'))*gendata(i,'Uini');
Parameter unit2(i,char);
unit2(i,'ch1')=24;unit2(i,'ch2')=
(gendata(i,'DT')-gendata(i,'S0'))*(1-gendata(i,'Uini'));
gendata(i,'Lj')=smin(char,unit(i,char));
gendata(i,'Fj')=smin(char,unit2(i,char)); Variable  costThermal   ;
Positive variables pu(i,t),p(i,t),StC(i,t),SDC(i,t),Pk(i,t,k),D(t);
Binary variable u(i,t),y(i,t),z(i,t);
p.up(i,t) = gendata(i,"Pmax") ;
p.lo(i,t) = 0; Pk.up(i,t,k)=data(k,i,'DP'); Pk.lo(i,t,k)=0;
p.up(i,t) = gendata(i,"Pmax") ; pu.up(i,h) = gendata(i,"Pmax") ;
Equations Uptime1,Uptime2,Uptime3,Dntime1,Dntime2,Dntime3,Ramp1,Ramp2,Ramp3,
Ramp4,startc ,shtdnc ,genconst1 ,Genconst2 ,Genconst3 ,Genconst4 ,balance ,DRconst;
Uptime1(i)$(gendata(i,"Lj")>0)..sum(t$(ord(t)< (gendata(i,"Lj")+1)),1-U(i,t))=e=0;
Uptime2(i)$(gendata(i,"UT")>1) ..
sum(t$(ord(t)>24-gendata(i,"UT")+1),U(i,t)-y(i,t))=g=0;
Uptime3(i,t)$(ord(t)>gendata(i,"Lj") and ord(t)<24-gendata(i,"UT")+2
and not(gendata(i,"Lj")>24-gendata(i,"UT"))    ) .. sum(h$((ord(h)>ord(t)-1) and
(ord(h)<ord(t)+gendata(i,"UT"))),U(i,h)) =g=gendata(i,"UT")*y(i,t);
Dntime1(i)$(gendata(i,"Fj")>0) .. sum(t$(ord(t)< (gendata(i,"Fj")+1)),U(i,t))=e=0;
Dntime2(i)$(gendata(i,"DT")>1) ..
sum(t$(ord(t)>24-gendata(i,"DT")+1),1-U(i,t)-z(i,t))=g=0;
Dntime3(i,t)$(ord(t)>gendata(i,"Fj") and ord(t)<24-gendata(i,"DT")+2 and
not(gendata(i,"Fj")>24-gendata(i,"DT"))    ) .. sum(h$((ord(h)>ord(t)-1) and
(ord(h)<ord(t)+gendata(i,"DT"))),1-U(i,h)) =g=gendata(i,"DT")*z(i,t);
startc(i,t) .. StC(i,t)=g=gendata(i,"costst")*y(i,t);
shtdnc(i,t) .. SDC(i,t)=g=gendata(i,"CostsD")*z(i,t);
```

```
genconst1(i,h)              .. p(i,h)=e=u(i,h)*gendata(i,"Pmin")+sum(k,Pk(i,h,k));
Genconst2(i,h)$(ord(h)>0) .. U(i,h)=e=U(i,h-1)$(ord(h)>1)
+gendata(i,"Uini")$(ord(h)=1)+y(i,h)-z(i,h);
Genconst3(i,t,k)       .. Pk(i,t,k)=l=U(i,t)*data(k,i,'DP');
Genconst4 ..costThermal=e=sum((i,t),StC(i,t)+SDC(i,t))
+sum((t,i),u(i,t)*gendata(i,'Mincost')+sum(k,data(k,i,'s')*pk(i,t,k)));
Ramp1(i,t) .. p(i,t-1)-p(i,t)=l=U(i,t)*gendata(i,'RD')+z(i,t)*gendata(i,"SD");
Ramp2(i,t) .. p(i,t)=l=pu(i,t);
Ramp3(i,t)$(ord(t)<24)..pu(i,t)=l=(u(i,t)-z(i,t+1))*gendata(i,"Pmax")
+z(i,t+1)*gendata(i,"SD");
Ramp4(i,t)$(ord(t)>1) ..pu(i,t)=l=p(i,t-1)+U(i,t-1)*gendata(i,'RU')
+y(i,t)*gendata(i,"SU");
balance(t) ..      sum(i,p(i,t))=e= D(t);
DRconst    ..      sum(t,dataLP(t,'load'))=e=sum(t,D(t));
Model UCLP /all/;
Option optcr=0.0; D.up(t)=1.1*dataLP(t,'load'); D.lo(t)=0.9*dataLP(t,'load');
Solve UCLP minimizing costThermal using mip ;
```

Table 5.6 Power schedules ($P_{i,t}$) by thermal units considering demand response constraint

Time	$g1$	$g3$	$g4$	$g5$	$g7$	$g8$
t_1			250.0	80.0	34.5	
t_2	92.9		250.0	80.0	30.0	
t_3	132.9		250.0	80.0	34.5	
t_4	146.0	80.0	250.0		34.5	38.0
t_5	146.0	110.0	250.0		34.5	38.0
t_6	146.0	120.0	250.0		34.5	38.0
t_7	146.0	120.0	250.0		34.5	38.0
t_8	146.0	120.0	250.0		34.5	38.0
t_9	146.0	120.8	250.0		34.5	38.0
t_{10}	146.0	120.8	250.0		34.5	38.0
t_{11}	146.0	120.0	250.0		34.5	38.0
t_{12}	146.0	120.0	250.0		34.5	38.0
t_{13}	146.0	120.0	250.0		34.5	38.0
t_{14}	146.0	120.0	250.0		34.5	38.0
t_{15}	146.0	120.0	250.0		34.5	38.0
t_{16}	146.0	120.0	250.0		34.5	38.0
t_{17}	146.0	120.0	250.0		34.5	38.0
t_{18}	144.0	115.0	250.0			38.0
t_{19}	128.8	105.0	250.0			30.0
t_{20}	128.0	96.7	250.0			
t_{21}	116.4	90.0	250.0			
t_{22}	110.0	78.1	250.0			
t_{23}	116.0	86.9	250.0			
t_{24}	97.8	65.0	250.0			

The demand pattern change in cost-based unit commitment with demand response flexibility is shown in Fig. 5.4. The demand flexibility is changed from 0 to 18% then the demand pattern changes are shown in Fig. 5.5. The variation of total operating costs vs demand flexibility is depicted in Fig. 5.6.



Fig. 5.4 The demand pattern change in cost-based unit commitment with demand response flexibility



Fig. 5.5 The sensitivity analysis of demand pattern changes in CBUC with various demand response flexibilities



Fig. 5.6 The variation of total operating costs vs demand flexibility

5.3 Price-Based Unit Commitment

The benefit maximization in price-based unit commitment is formulated in (5.12).

$$\max_{p^k_{i,t},u_{i,t},y_{i,t},z_{i,t}} \quad \text{OF} = \sum_{i,t} \{\lambda_t p_{i,t} - [\text{FC}_{i,t} + \text{STC}_{i,t} + \text{SDC}_{i,t}]\} \tag{5.12}$$

Subject to:

(5.2), (5.3), (5.4), (5.5), (5.6), (5.7)

The GAMS code for solving the (5.12) is described in GCode 5.5.

GCode 5.5 Price-based unit commitment Example for ten unit system

```
Sets t   /t1*t24/, i /g1*g10 /,k /sg1*sg20/,char /ch1*ch2/;
Alias (t,h)
Table Gdata(i,*) generator cost characteristics and limits
     a  b c  CostsD  costst RU RD UT DT SD SU Pmin Pmax U0 Uini S0
* Removed for saving space
Parameter data(k,i,*);
data(k,i,'DP')=(Gdata(i,"Pmax")-Gdata(i,"Pmin"))/card(k);
data(k,i,'Pini')= (ord(k)-1)*data(k,i,'DP')+Gdata(i,"Pmin");
data(k,i,'Pfin')=data(k,i,'Pini')+data(k,i,'DP');
data(k,i,'Cini')=Gdata(i,"a")*power(data(k,i,'Pini'),2)
```

```
                         +Gdata(i,"b")*data(k,i,'Pini')+Gdata(i,"c");
data(k,i,'Cfin')=Gdata(i,"a")*power(data(k,i,'Pfin'),2)
                         +Gdata(i,"b")*data(k,i,'Pfin')+Gdata(i,"c");
data(k,i,'s')= (data(k,i,'Cfin')-data(k,i,'Cini'))/data(k,i,'DP');
Gdata(i,'Mincost')=Gdata(i,'a')*power(Gdata(i,"Pmin"),2)
                         +Gdata(i,'b')*Gdata(i,"Pmin") +Gdata(i,'c');
table dataLP(t,*)
          lambda        load
* Removed for saving space ;
parameter unit(i,char); unit(i,'ch1')=24;
unit(i,'ch2')=(Gdata(i,'UT')-Gdata(i,'U0'))*Gdata(i,'Uini');
Parameter unit2(i,char);
unit2(i,'ch1')=24;
unit2(i,'ch2')=(Gdata(i,'DT')-Gdata(i,'S0'))*(1-Gdata(i,'Uini'));
Gdata(i,'Lj')=smin(char,unit(i,char));
Gdata(i,'Fj')=smin(char,unit2(i,char));Variable  costThermal,OF;
Positive variables pu(i,t),p(i,t),StC(i,t),SDC(i,t),Pk(i,t,k);
binary variable u(i,t),y(i,t),z(i,t);
p.up(i,t) = Gdata(i,"Pmax") ; p.lo(i,t) = 0;
Pk.up(i,t,k)=data(k,i,'DP'); Pk.lo(i,t,k)=0;
p.up(i,t) = Gdata(i,"Pmax") ; pu.up(i,h) = Gdata(i,"Pmax") ;
Eequations Upt1,Upt2,Upt3,Dntime1,Dntime2,Dntime3,Ramp1,Ramp2,Ramp3,Ramp4,
startc,shtdnc,genconst1, Genconst2,Genconst3,Genconst4,balance,benefitcalc;
Upt1(i)$(Gdata(i,"Lj")>0)..sum(t$(ord(t)< (Gdata(i,"Lj")+1)),1-U(i,t))=e=0;
Upt2(i)$(Gdata(i,"UT")>1)..sum(t$(ord(t)>24-Gdata(i,"UT")+1),U(i,t)-y(i,t))=g=0;
Upt3(i,t)$(ord(t)>Gdata(i,"Lj") and ord(t)<24-Gdata(i,"UT")+2 and
not(Gdata(i,"Lj")>24-Gdata(i,"UT"))) ..sum(h$((ord(h)>ord(t)-1) and
(ord(h)<ord(t)+Gdata(i,"UT"))),U(i,h)) =g=Gdata(i,"UT")*y(i,t);
Dntime1(i)$(Gdata(i,"Fj")>0)..sum(t$(ord(t)< (Gdata(i,"Fj")+1)),U(i,t))=e=0;
Dntime2(i)$(Gdata(i,"DT")>1)..
 sum(t$(ord(t)>24-Gdata(i,"DT")+1),1-U(i,t)-z(i,t))=g=0;
Dntime3(i,t)$(ord(t)>Gdata(i,"Fj") and ord(t)<24-Gdata(i,"DT")+2 and
not(Gdata(i,"Fj")>24-Gdata(i,"DT"))) .. sum(h$((ord(h)>ord(t)-1) and
(ord(h)<ord(t)+Gdata(i,"DT"))),1-U(i,h)) =g=Gdata(i,"DT")*z(i,t);
startc(i,t) .. StC(i,t)=g=Gdata(i,"costst")*y(i,t);
shtdnc(i,t) .. SDC(i,t)=g=Gdata(i,"CostsD")*z(i,t);
genconst1(i,h) .. p(i,h)=e=u(i,h)*Gdata(i,"Pmin")+sum(k,Pk(i,h,k));
Genconst2(i,h)$(ord(h)>0) .. U(i,h)=e=U(i,h-1)$(ord(h)>1)
                                       +Gdata(i,"Uini")$(ord(h)=1)+y(i,h)-z(i,h);
Genconst3(i,t,k) .. Pk(i,t,k)=l=U(i,t)*data(k,i,'DP');
Genconst4 .. costThermal=e=sum((i,t),StC(i,t)+SDC(i,t))
                                        +sum((t,i), u(i,t)*Gdata(i,'Mincost')
                                        +sum(k,data(k,i,'s')*pk(i,t,k)));
Ramp1(i,t) .. p(i,t-1)-p(i,t)=l=U(i,t)*Gdata(i,'RD')+z(i,t)*Gdata(i,"SD");
Ramp2(i,t) .. p(i,t)=l=pu(i,t);
Ramp3(i,t)$(ord(t)<24) .. pu(i,t)=l=(u(i,t)-z(i,t+1))*Gdata(i,"Pmax")
                                                 +z(i,t+1)*Gdata(i,"SD");
Ramp4(i,t)$(ord(t)>1)  .. pu(i,t)=l=p(i,t-1)+U(i,t-1)*Gdata(i,'RU')
                                                +y(i,t)*Gdata(i,"SU");
Balance(t)  ..  sum(i,p(i,t))=l= dataLP(t,'load');
Benefitcalc ..  OF=e=sum((i,t),dataLP(t,'lambda')*p(i,t))- costThermal
Model UCLP /all/;
Option optcr=0.0;
Solve UCLP maximizing OF using mip ;
```



Fig. 5.7 Power schedules in price-based unit commitment vs time

The start-up, shut-down, and fuel costs are $499.000, $163.800, and $372,277.510, respectively. The net benefit of Genco is $58,186.325. The power schedules of thermal units in cost-based unit commitment are depicted in Fig. 5.7.

5.4 Applications

The unit commitment analysis has a vast range of applications, and some of them are described as follows:

5.4.1 Cost-Based UC

In practical applications, the thermal units might have some limitations in fuels. This means that the unit cannot generate more than some certain amount of energy. In other words, it might not be technically possible to keep the unit on for all time steps for the operating period. The fuel constrained UC is formulated in 4,335,178. Another important constraint in UC is environmental emission which is modeled in [4].

Another aspect of UC is considering the uncertainties associated with renewable energy resources (like wind turbines). Generally speaking, there are several methods for dealing with uncertainties in power system studies, such as

- Stochastic methods: two-stage scenario-based [5], two-point estimate [6, 12]
- Fuzzy methods [7]
- Robust optimization [8, 9]
- Information gap decision theory [10, 11]

The different uncertainty parameters and modeling methods are shown in Fig. 5.8 [12].



Fig. 5.8 Different uncertainty parameters and modeling methods

5.4.2 Price-Based UC

Some price-based UC formulations are listed as follows:

- Robust optimization-based self-scheduling of hydro-thermal Genco in smart grids [13]
- Risk averse optimal operation of a virtual power plant using two-stage stochastic programming [14]
- Smart self-scheduling of Gencos with thermal and energy storage units under price uncertainty [15].

References

1. A. Ademovic, S. Bisanovic, M. Hajro, A genetic algorithm solution to the unit commitment problem based on real-coded chromosomes and fuzzy optimization, in *Melecon 2010 - 2010 15th IEEE Mediterranean Electrotechnical Conference*, April 2010, pp. 1476–1481
2. J.M. Arroyo, A.J. Conejo, Optimal response of a thermal unit to an electricity spot market. IEEE Trans. Power Syst. **15**(3), 1098–1104 (2000)
3. M. Ortega-Vazquez, D. Kirschen, Optimizing the spinning reserve requirements using a cost/benefit analysis, in *2008 IEEE Power and Energy Society General Meeting - Conversion and Delivery of Electrical Energy in the 21st Century*, July 2008, p. 1

4. T. Gjengedal, Emission constrained unit-commitment (ECUC). IEEE Trans. Energy Convers. **11**(1), 132–138 (1996)
5. C. Uçkun, A. Botterud, J.R. Birge, An improved stochastic unit commitment formulation to accommodate wind uncertainty. IEEE Trans. Power Syst. **31**(4), 2507–2517 (2016)
6. A. Soroudi, M. Aien, M. Ehsan, A probabilistic modeling of photo voltaic modules and wind power generation impact on distribution networks. IEEE Syst. J. **6**(2), 254–259 (2012)
7. A. Soroudi, Possibilistic-scenario model for DG impact assessment on distribution networks in an uncertain environment. IEEE Trans. Power Syst. **27**(3), 1283–1293 (2012)
8. R. Jiang, J. Wang, Y. Guan, Robust unit commitment with wind power and pumped storage hydro. IEEE Trans. Power Syst. **27**(2), 800–810 (2012)
9. A. Soroudi, P. Siano, A. Keane, Optimal DR and ESS scheduling for distribution losses payments minimization under electricity price uncertainty. IEEE Trans. Smart Grid **7**(1), 261–272 (2016)
10. A. Soroudi, A. Rabiee, A. Keane, Information gap decision theory approach to deal with wind power uncertainty in unit commitment. Electr. Power Syst. Res. **145**, 137–148 (2017)
11. C. Murphy, A. Soroudi, A. Keane, Information gap decision theory-based congestion and voltage management in the presence of uncertain wind power. IEEE Trans. Sust. Energy **7**(2), 841–849 (2016)
12. A. Soroudi, T. Amraee, Decision making under uncertainty in energy systems: state of the art. Renew. Sust. Energ. Rev. **28**, 376–384 (2013)
13. A. Soroudi, Robust optimization based self scheduling of hydro-thermal Genco in smart grids. Energy **61**, 262–271 (2013)
14. M.A. Tajeddini, A. Rahimi-Kian, A. Soroudi, Risk averse optimal operation of a virtual power plant using two stage stochastic programming. Energy **73**, 958–967 (2014)
15. A. Soroudi, Smart self-scheduling of Gencos with thermal and energy storage units under price uncertainty. Int. Trans. Electr. Energy Syst. **24**(10), 1401–1418 (2014)